
Abstract

We approach the general problem of representing higher-order languages, that are
usually equipped with special variable binding constructs, in a less specialized first-
order framework such as membership equational logic and the corresponding version
of rewriting logic. The solution we propose is based on CINNI, a new calculus of
explicit substitutions that makes use of a term representation that contains both
the standard named notation and de Bruijn’s indexed notation as special subcases.
The calculus is parametric in the syntax of the object language, which allows us to
apply it to different object languages such as λ-calculus, Abadi and Cardelli’s object
calculus (ς-calculus) and Milner’s calculus of communicating mobile processes (π-
calculus). As a practical result we obtain executable formal representations of these
object languages in Maude with a representational distance close to zero.

1 Introduction

A common feature of higher-order languages is that essential entities which
operate on data receive a first-class status so that variables can range over
these entities. For instance, in higher-order logics or higher-order functional
programming languages, variables can range over functions; in object-oriented
programming languages, variables can range over objects (which can contain
methods); and in languages for mobile processes, variables can range over
processes or channels (which are references to processes). In order to express
the essential entities directly as terms in the language, higher-order languages
typically provide a syntax for abstractions, such as abstractions for functions,

methods or processes in the examples above. Abstractions are essentially
binding constructs that bind the free variables in the abstracted term with
the intention to be instantiated later by means of substitution.

Currently there are two major approaches to representing an object language
with binding constructs in first-order frameworks such as membership equa-
tional logic [28,10] or rewriting logic [27]. We distinguish betweeen represen-
tations with names and representations based on de Bruijn indices. Represen-
tations with names have the very desirable feature to be close to the object
language, but a major drawback is that they lack a canonical way to treat
names and potential name clashes. Calculi based on de Bruijn indices have
the advantage of a canonical representation, but they are more abstract, since
information about names is not represented so that the gap between the object
language and its representation is considerable.

In addition to the representation of the object language, another important
issue is capture-free substitution, the main operation that is needed for terms
of the object language. For a definition of substitution on terms with binding
constructs, it turns out to be useful to treat substitutions as first-class citizens,
since substitutions usually have to be adjusted as they are propagated through
the term they are applied to. Since substitutions receive the same formal
status as terms, the calculi that deal with substitution in this way are called
explicit substitution calculi.

Research in this area has led to a rich collection of calculi (overviews and com-
parisons can be found in [25,35,8,30,22]) with quite different properties and
motivations. Most of this research is focused on λ-calculus (notable exceptions
are [9] and [33]) and the use of explicit substitutions to express β-reduction in
terms of a more primitive concept. Among the motivations for using explicit
substitutions we can find the following: the need for a rigorous and simple
explanation of capture-free substitution [4], the quest for a notion of com-
putation that is more fine-grained and more implementation-oriented than
standard β-reduction [14,2], the interest in analysis of evaluation strategies
and efficiency of computation [19], the application of first-order techniques to
higher-order languages [18], and the use of algorithms that operate on incom-
plete terms such as higher-order unification [17], type checking/inference [36],
and proof synthesis [31,30].

Our primary motivation to propose a new calculus of explicit substitutions in
this paper is to obtain a first-order representation of terms with binding and
capture-free substitution that is as close as possible to the standard named
notation. Beyond that, we want this calculus to be executable, in the sense
that it can be executed using a rewriting engine such as Maude [12,11], and
furthermore we are interested in a general solution, in the sense that the the-
ory does not restrict ourselves to a particular object language. More precisely,
the objective of this paper is to develop a calculus of names and explicit sub-
stitutions that takes names seriously and completely removes the gap between
the object language and its representation (often called representational dis-

tance) without losing the possibility of canonical representations. A solution
that is closely related to de Bruijn’s representation [16] but has been devel-
oped independently by Berkling [6,7] is a unification of named and indexed
notation. Despite of its advantages, which have also been recognized more
recently in [34], it is an unconventional representation that has not attracted
much attention so far. Therefore, we devote the next section to an introduc-
tion and motivation of Berkling’s representation, which will serve as a basis
for the CINNI substitution calculus.

2 Indexed Names and Named Indices

Consider the standard treatment of binding constructs, say in the context of
first-order logic, where α-equivalent terms, i.e. terms that can be transformed
into each other by consistent renaming of bound variables, are identified, i.e.
not distinguished for essential parts of the metatheory. An obvious first step
towards a named representation is to give up this identification that we also
refer to as α-equality. Unfortunately, this rather naive approach leads to the
following difficulty that we refer to as accidental hiding.

Consider for instance the formula

∀X.(A ∧ ∀Y.(B ⇒ ∀X.C(X)))

for distinct names X and Y . Assume the subformula C(X) contains X free.
Then each free occurrence of X in C(X) is captured by the inner ∀ quanti-
fier, so that the name bound by the outermost ∀ quantifier is hidden from
the viewpoint of C(X). Indeed there is no way to refer to the outermost ∀
quantifier within C(X).

Hence, we are faced with the following problem: a calculus without α-equality
is not only less abstract, which is an unavoidable consequence of giving up
identification by α-conversion, but also, depending on the (accidental) choice
of names, visibility of (bound) variables may be restricted. It is important
to emphasize that visibility is not restricted in the original calculus with α-
equality, since renaming can be performed tacitly at any time.

Clearly, the phenomenon of hiding that occurs in the example above is undesirable 1 ,
because it is not present in the original calculus with α-equality. It is merely
an accident caused by giving up identification by α-conversion without adding
a compensating flexibility to the language.

This suggests tackling this general problem by migrating to a more flexible
syntax, where we express a binding constraint by annotating each name X
with an index i ∈ N, written Xi, that indicates how many X-binders should
be skipped before we reach the one that Xi refers to. For instance, we write

∀X.(A ∧ ∀Y.(B ⇒ ∀X.C(X0)))

1 Of course, in general hiding is important but it is not an issue of binding; it should be
treated independently.

to express that X0 is bound by the inner ∀, and

∀X.(A ∧ ∀Y.(B ⇒ ∀X.C(X1)))

meaning that X1 is bound by the outermost ∀. To make the language a
conservative extension of the traditional notation, we can identify X and X0.

The use of indexed names is equivalent to a representation introduced by
Berkling [6,7] in the context of λ-calculus 2 which is why we refer to the
notation based on indexed names also as Berkling’s notation. As indicated by
the example above we use Berkling’s representation not (only) for λ-calculus
but as the core syntax of CINNI, the Calculus of Indexed Names and Named
Indices which is generic in the sense that it can be instantiated for a wide
range of object languages.

Obviously, there is some similarity to a notation based on de Bruijn indices
[16], but notice that there is an essential difference: the index m in the oc-
currence Xm is not the number of binders to be skipped; it states that we
have to skip m binders for the particular name X, not counting binders for
other names. Still a formal relationship to de Bruijn’s notation can be estab-
lished: if we restrict ourselves to terms that contain only a single name X,
then we can replace each Xi by the index i without loss of information and we
arrive at de Bruijn’s purely indexed notation. 3 In other words, if we restrict
the available names to a single one, we obtain de Bruijn’s notation as a very
special case. In this sense, Berkling’s representation can be formally seen as
a proper generalization of de Bruijn’s notation. Pragmatically, however, the
relationship to de Bruijn’s syntax plays only a minor role, since a typical user
will exploit the dimension of names much more than the dimension of indices.
Hence, in practice the notation can be used as a standard named notation,
with the additional advantage that accidental hiding and weird renamings 4

are avoided.

The pragmatic advantage of Berkling’s notation is that it can be used to
reduce the distance between the formal system and its implementation: it can
be directly employed by the user who wants to think in terms of names, so
that the need for a translation between an internal representation (e.g., using
de Bruijn indices) and a user-friendly syntax (e.g., using ordinary names)
disappears completely.

Usually, this translation between an internal and an external representation
is not considered to be a problem, and indeed, in the case of terms where all
parts are known or accessible, solutions are straightforward. However, even
in this case this gap is not desirable; consider, for example, a tactic-based
theorem prover where the user is confronted with an internal representation
which reflects the theory only in a very indirect way. More seriously, the

2 An indexed variable Xi is represented in Berkling’s representation as #iX where # is
the so-called unbinding operation.
3 With the slight difference that de Bruijn’s indices start at 1 instead of 0.
4 See the discussion on weird renaming in the next section.

translation between internal and external representations becomes impossible,
or at least requires certain restrictions, as soon as we use terms containing
metavariables, holes or placeholders, which are useful for many applications
including unification algorithms and representation of incomplete proofs.

3 Explicit Substitutions

In the previous section we discussed Berkling’s first-order representation for
expressions which contains the conventional named notation as well as de
Bruijn’s indexed notation as special cases. The most important operation to
be performed on such terms represented in this way is capture-free substitu-
tion. Therefore, we now present the CINNI substitution calculus, a first-order
calculus that can be seen as an (operational) refinement of an external (i.e.
metalevel) substitution function such as the one given in [7].

Strictly speaking, CINNI is a family of explicit substitution calculi, parame-
terized by the syntax of the language we want to represent. For a language
L given by its syntax we denote the corresponding instantiation of CINNI by
CINNIL. The syntax defines the term constructors together with their binding
constraints which are expressed by associating a binary relation to each term
constructor f as follows: We say that f binds argument i in argument j iff for
each term f(P1, . . . , Pn) of the object language, Pi is a name and this name is
bound in the subterm Pj. So each Pi is either a term or a name (names are
not considered to be object language terms).

As an example we use the untyped λ-calculus to present the concrete instan-
tiation CINNIλ of the substitution calculus. CINNIλ-terms are generated by
the syntax

Xm | (M N) | [X]M

with the constraint that [ ] binds argument 0 in argument 1.

As a motivation for the substitution calculus given below, consider the follow-
ing example of a β-reduction step in the traditional λ-calculus with distinct
names X and Y , again taking names literally, i.e., not presupposing identifi-
cation by α-conversion:

(([X][Y ]X)Y ) →β [Z]Y

Clearly, the bound variable Y must be renamed to Z, a name different from Y ,
to avoid capturing of the free variable Y . Unfortunately, there is no canonical
choice if all names should be treated as being equal. We call this phenomenon
weird renaming of bound variables. It is actually a combination of two un-
desirable effects: (1) names that have been carefully chosen by the user have
to be changed, and (2) the enforced choice of a new name collides with the
right of names to be treated as equal citizens. These effects are avoided in
the CINNI calculus. It is specified by the first-order equational theory given
below. Indeed, the only operation assumed on names is equality.

More formally, we assume that the syntax of the object language is given

by a signature in membership equational logic which introduces a sort of
names (assumed to be nonempty), a sort of natural numbers (with zero 0 and
successor +1 as the only operations), and a sort of object language terms. In
addition, we have a constructor for variables, i.e. terms of the form Xm, as well
as additional constructors for terms together with their binding constraints
and optional structural equations (see below).

To present the actual calculus we need to extend the notion of term by explicit
substitutions. To this end, we introduce a sort of substitutions together with
the following operators: In addition to the two basic kinds of substitutions,
namely simple substitutions [X:=M ] and shift substitutions ↑X, substitutions
can be lifted using ⇑X (S), where the variable S ranges over substitutions. The
application S M of an explicit substitution to a term is again a term. Now
CINNIL has the signature just described and the following equations:

[X:=M ] X0 = M (FVar)

[X:=M ] Xm+1 = Xm (RVarEq)

[X:=M ] Yn = Yn if X 
= Y (RVarNEq)

↑XXm = Xm+1 (VarShiftEq)

↑XYn = Yn if X 
= Y (VarShiftNEq)

⇑X (S) X0 = X0 (FVarLift)

⇑X (S) Xm+1 = ↑X (S Xm) (RVarLiftEq)

⇑X (S) Yn = ↑X (S Yn) if X 
= Y (RVarLiftNEq)

For each syntactic constructor f of L we add a syntax-specific equation

S f(P1, . . . , Pn) = f(⇑Pj1,1
(. . . ⇑Pj1,m1

(S)) P1, . . . ,⇑Pjn,1
(. . . ⇑Pjn,mn

(S)) Pn)

where ji,1, . . . , ji,mi
are all the arguments (necessarily of the name sort) that

f binds in argument i (necessarily of term sort). If Pk is of name sort we
identify S Pk and Pk (abuse of notation).

Often the terms of the object language are not just freely generated by the
syntactic constructors, but are subject to additional structural equations. Ad-
missible equations in this paper are the laws of associativity, commutativity
and identity for binary operators, and we assume throughout the paper that
structurally equivalent terms are identified. If f is a binary operator with
identity e we add a condition to the syntax-specific equation above which en-
sures that none of the Pi is equal to e (cf. the specifications of ς-calculus
and π-calculus in Section 5). This ensures that the left hand sides of valid
instances of syntax-specific equations do not overlap and is also needed to
ensure termination of the corresponding rewrite system.

The syntax-specific equations are the only equations that depend on the syntax

of L. For instance, CINNIλ has the following syntax-specific equations:

S (MN) = (SM)(SN) (App)

S ([X]M) = [X](⇑X (S) M) (Lambda)

The equations of CINNI can be justified by the following algebraic substitu-
tion semantics: a substitution S is interpreted as a function from variables
to terms. Application of substitution is interpreted as function application.
The substitutions [X:=M ], ↑X and ⇑X (S) are then uniquely defined by the
equations above. Finally, substitutions are extended from variables to terms
by the syntax-specific equations. Each time a substitution moves into a new
scope it has to be adjusted using a lift substitution.

CINNI is not only an equational calculus with an algebraic semantics, but as
usual for explicit substitution calculi it can be equipped with an operational
semantics by regarding equations as rewrite rules. We refer to the result-
ing term rewrite system as the CINNI rewrite system and we introduce the
following relations: The characteristic relation of the equations of CINNI is
denoted by ⇒S, i.e. M ⇒S N holds iff M = N is a valid instance of one of the
equations. The rewrite relation induced by ⇒S, i.e. its compatible closure, is
denoted by →S. The induced equivalence on terms is denoted by =S.

We can now define the explicit substitution version of the β-rule by

([X]N)M ⇒B [X:=M ]N.

Notice that weird renaming of bound variables as in the previous example is
avoided with the new notion of β-reduction. For instance, we have

(([X][Y ]X0)Y0) →∗
SB ([Y ]Y1)

where →SB denotes the compatible closure of ⇒S ∪ ⇒B. Notice also that we
do not view application-specific computation rules as a part of the substitu-
tion calculus (which is CINNIλ in this case). The substitution calculus only
depends on the syntax of the object language.

As another application of substitution, consider renaming of a bound variable
X by • as in the following explicit substitution version of α-reduction

([X]N) ⇒A ([ • ][X:= • ] ↑•N) if X 
= •

where • is an arbitrary but fixed name. Using this rule and the rules for explicit
substitutions, every CINNIλ term can be reduced to a nameless α-normal form
which is essentially its de Bruijn index representation. α-reduction is a new
concept that becomes expressible due to the use of a unified syntax with
indices and names.

Just as Berkling’s notation contains de Bruijn’s notation as a very special
case, the instantiation of CINNI for the λ-calculus reduces to the calculus

7



Stehr

λυ of explicit substitutions proposed by Pierre Lescanne [25,26,4], but only
in the degenerate case where we restrict the set of names to a singleton set.
It is noteworthy that λυ is the smallest known indexed substitution calculus
enjoying good theoretical properties like confluence and preservation of strong
normalization. It seems that its simplicity is inherited by CINNI although
in practice the dimension of names will be much more important than the
dimension of indices. Hence, we tend to think of CINNI more as a substitution
calculus with names than as one with indices.

4 Metatheoretic Properties of CINNI

In this secion we give a number of important operational properties concerning
both the CINNI calculus in isolation, and the composition of CINNI with
application-specific rules such as the explicit substitution version of the β-rule
in the case of CINNIλ. The present section generalizes the results of [25] for
λυ in two orthogonal dimensions.

The first dimension is the scope of applicability:

(i) Instead of considering a fixed object language such as λ-calculus, we
consider an arbitrary object language given by a syntax L with binders.

(ii) Instead of considering a fixed set of computation rules, such as β-reduction
in λ-calculus, we allow arbitrary computation rules R as long as they sat-
isfy a certain well-formedness property.

The second dimension of generalization is concerned with the representation
of the object language:

(i) The de Bruijn index representation is generalized to a richer representa-
tion with indexed names.

(ii) The substitution calculus and its properties are generalized accordingly.

Due to space limitations proofs of the metatheoretic properties cannot be
given in this section, but detailed proofs can be found in the extended version
of this paper [37].

4.1 CINNI in Isolation

We first consider the CINNI rewrite system in isolation, instantiated for the
syntax L of an arbitrary object language.

Definition 4.1 We define the following two functions on terms to express
iterated shift and lift substitutions:

↑∅(M) = M

↑Y ,X (M) =↑Y (↑XM)

⇑∅(S)(M) = S M

⇑Y ,X (S)(M) =⇑Y (⇑X (S))(M)

Here and in the following we use variables X, Y , Z, U , V , W to range over
lists of names, ∅ denotes the empty list and comma denotes concatenation.
Furthermore, we use |Y |X to denote the number of occurrences of X in Y .

4.1.1 Operational Properties

Confluence of CINNI can be easily established by transforming the rewrite
system into an equivalent orthogonal, i.e. left-linear and non-overlapping,
rewrite system without conditions. This is done by replacing each possibly
conditional equation by all its valid instances obtained by instantiating X and
Y by concrete names. Notice that the resulting system becomes infinite if the
set of names is infinite.

Theorem 4.2 The relation →S is confluent.

Some mathematical evidence that CINNIλ is a nontrivial generalization of
λυ and its metatheory seems to be given by the observation that the proof
of strong normalization in [4,26], which makes use of elementary interpreta-
tions [24], cannot be applied to CINNIλ. Indeed the problematic equations are
(RVarLiftNEq) and the syntax-specific equations which make it unlikely that a
proof of strong normalization can be obtained by a modified elementary inter-
pretation. Furthermore, the syntax-specific equations seem to prevent us from
giving a proof based on recursive path orderings. Hence, we pursue a different
approach which makes use of the fact that in orthogonal term rewrite systems
all maximal computations that do not erase redices are essentially equivalent.
In particular, computations that only perform innermost reductions are in a
certain sense representative computations and can be used to prove strong
normalization as already observed in [32]. Hence the following theorem can
be proved by exhibiting an innermost-reduction strategy and showing that it
always terminates.

Theorem 4.3 The relation →S is strongly normalizing.

As a consequence of this theorem, each CINNIL-term M and each CINNIL-
substitution S has a unique substitution normal form which will be denoted
by NFS(M) and NFS(S), respectively. Notice that NFS(M) does not contain
any substitutions, otherwise one of the rules could be applied.

4.1.2 Equational Properties

The following induction lemma provides a tool for proving certain equivalences
of the form SL

l . . . SL
1 M =S SR

r . . . SR
1 M , and it has been used in the

proofs of the subsequent lemmas which state basic equational properties of
the CINNI-calculus.

Lemma 4.4 (Induction Lemma)

Let SL
1 . . . SL

l and S
R
1 . . . SR

r be substitutions. Define

L(M) = SL
l . . . SL

1 M, ⇑W (L)(M) = ⇑W (SL
l ) . . . ⇑W (SL

1 ) M,

R(M) = SR
r . . . SR

1 M, ⇑W (R)(M) = ⇑W (SR
r ) . . . ⇑W (SR

1 ) M.

In order to prove

⇑W (L)(M) =S ⇑W (R)(M) and in particular L(M) =S R(M)

for all M , W , it is sufficient to show that

⇑W (L)(Xk) =S ⇑W (R)(Xk)

for all X, k, and W .

Lemma 4.5 (Simplification I)

(i) ⇑Y (S) Xm =S Xm if m < |Y |X ,
(ii) ⇑Y (S) Xm =S ↑Y S Xm−i if m ≥ |Y |X = i.

Lemma 4.6 (Simplification II)

(i) ⇑Y (↑Z) Xm =S Xm where Z 
= X,

(ii) ⇑Y (↑Z) Xm =S ↑Z Xm if m ≥ |Y |X ,
(iii) ⇑Y (↑Z) Xm =S ↑Z Xm if Z /∈ Y ,

(iv) ⇑W (⇑Y (↑Z)) M =S ⇑W (↑Z) M if Z /∈ Y ,

(v) ⇑Y (↑Z) M =S ↑Z M if Z /∈ Y .

Lemma 4.7 (Simplification III)

(i) ⇑Y ([Z:=N ]) Xm =S Xm where Z 
= X,

(ii) ⇑Y ([Z:=N ]) ⇑Y (↑Z) Xm =S Xm,

(iii) ⇑Y ([Z:=N ]) ⇑Y (↑Z) M =S M .

In each of the following lemmas we explicitly state a strong equivalence, that
can be established using the induction lemma, followed by weaker conse-
quences that are typically sufficient for most purposes.

Lemma 4.8 (Shift Shift Reordering)

(i) ⇑W (↑Z) ⇑W (↑Y) M =S ⇑W (↑Y) ⇑W (↑Z) M ,

(ii) ↑Z ↑Y M =S ↑Y ↑Z M .

Lemma 4.9 (Lift Lift Reordering)

(i) ⇑W (⇑Z (⇑Y (S))) M =S ⇑W (⇑Y (⇑Z (S))) M ,

(ii) ⇑Z (⇑Y (S)) M =S ⇑Y (⇑Z (S)) M .

Lemma 4.10 (General Shift Reordering)

(i) ⇑W,Y (S) ⇑W (↑Y) M =S ⇑W (↑Y) ⇑W (S) M ,

(ii) ⇑Y (S) ↑Y M =S ↑Y S M ,

(iii) ⇑Y (S) ↑Y M =S ↑Y S M .

Lemma 4.11 (Simple Substitution Reordering)

(i) ⇑W (S) ⇑W ([Z:=N ]) M =S ⇑W ([Z:=S N ]) ⇑W,Z (S) M ,

(ii) S [Z:=N ] M =S [Z:=S N ] ⇑Z (S) M ,

(iii) [Y :=L] [Z:=N ] M =S [Z:=[Y :=L] N ] ⇑Z ([Y :=L]) M .

4.2 Preservation of Confluence

In a typical application context CINNIL is extended by extra equations such
as β-reduction in the case of CINNIλ and a natural question is whether the
resulting system remains confluent if confluence of the system without explicit
substitutions has already been established. So let R be the set of (possibly
conditional) equations of the form M = N if C. Here M and N are terms
possibly containing equational logic variables. Assume furthermore that ⇒R

is the characteristic relation defined by these extra equations on terms. We
denote the compatible closures of ⇒R and ⇒S ∪ ⇒R on terms and substi-
tutions by →R and →SR, respectively. We also define a relation �R on pure,
i.e. substitution-free, terms by M �R M

′′ iff there is an M ′ such that M →R

M ′ and M ′′ = NFS(M
′). We use �R as a reference system that operates on

a level of abstraction without (observable) explicit substitutions.

Definition 4.12 (Well-Formedness)

We say that the set R of equations is well-formed iff for each equation in
R both sides of are terms of the (object language) term sort, there are no
substitution applications occurring in the left hand side, and for all terms
M ,N , and substitutions S,

M ⇒R M
′ implies ∃N ′ : NFS(S M) ⇒R N

′ =S S M ′

Intuitively, well-formedness expresses compatibility between application of
rules and application of substitutions modulo the equations of the substitution
calculus. It is interesting to note that well-formedness excludes quite different
kinds of ill-formed rules. For a concrete object language well-formedness can
typically be verified using the lemmas given before.

Lemma 4.13 (Well-Formedness Lemma) Assume that R is well-formed.
Then for all terms M ,N , n ∈ N, and substitutions S1, . . . , Sn,

M ⇒R M
′ implies ∃N ′ : NFS(Sn . . . S1 M) ⇒R N

′ =S Sn . . . S1 M
′

It is remarkable that the proof of the confluence theorem and in particular
the proof of the subsequent projection lemma, both given in [26,4] for λυ,
generalize to our setting without any difficulties (see [37]).

Lemma 4.14 (Projection) Assume that R is well-formed. Then

(i) M →R M
′ implies NFS(M) �∗

R NFS(M
′) and

(ii) S →R S
′ implies NFS(S) �∗

R NFS(S
′).

Lemma 4.15 (Hardin’s Interpretation Technique [2,20])

Let S and R be relations on some set T . Assume S is confluent and terminat-
ing and NFS(x) is the normal form of x w.r.t. S. Assume RS is a relation on

NFS(T ) with RS ⊆ (S ∪R)∗ and x R y implies NFS(x) RS
∗ NFS(y). Then

confluence of RS implies confluence of (S ∪R)∗.

As a direct consequence of the previous two lemmas we obtain:

Theorem 4.16 (Preservation of Confluence)

If R is well-formed and �R is confluent then →SR is confluent.

A noteworthy point is that confluence is not reduced to local confluence via
Newman’s Lemma. Therefore the previous theorem can also be applied in
cases where �R is not strongly normalizing.

5 Applications

In this section we illustrate the use of CINNI to obtain membership equa-
tional logic [28,10] specifications of λ-calculus, Abadi and Cardelli’s ς-calculus
[1] as well as a rewriting logic [27] specification of Milner’s π-calculus [29].
These calculi are interesting, since they have different binding constructs and
quite different equational theories. In addition, the π-calculus does not only
have an equational theory that specifies process congruence, but there are
also rewrite rules to specify the operational semantics in terms of a transi-
tion system equipped with an algebraic structure. Both process congruence
equations and transition rules make use of substitutions. We show how appro-
priate instantiations of CINNI can be used in all three cases to obtain formal
and executable first-order representations of these languages in Maude [12,11].
Furthermore, in each of these examples we give application-specific confluence
results that can be obtained using the general results stated earlier. In each
example the well-formedness condition can be verified using the lemmas in
Section 4.1.2. Since we aim at a unique model in each case, the specifications
we give in the following should all be interpreted under the initial semantics.

5.1 Higher-Order Functions: Lambda-Calculus

For the representation of untyped λ-calculus we use the predefined sort Qid to
represent names. The following signature defines representations of variables
and λ-terms as elements of the sorts Var and Trm, respectively:

sorts Var Trm .

op _{_} : Qid Nat -> Var .

subsort Var < Trm .

op __ : Trm Trm -> Trm .

op [_]_ : Qid Trm -> Trm .

vars n m : Nat . vars X Y Z : Qid . vars M N : Trm .

Here X{m} is the representation of a variable, i.e. an indexed name, while
(M N) and [X] M represent application and abstraction, respectively.

12

The instantiation of CINNI to the syntax of λ-terms, that is CINNIλ, is given
below. [X := M], [shift X], and [lift X S] represent simple substitutions,
shift substitutions, and lifted substitutions, respectively, and __ is substitution
application.

sort Subst . var S : Subst .

op [_:=_] : Qid Trm -> Subst .

op [shift_] : Qid -> Subst .

op [lift__] : Qid Subst -> Subst .

op __ : Subst Trm -> Trm .

eq ([X := M] (X{0})) = M .

eq ([X := M] (X{suc(m)})) = (X{m}) .

ceq ([X := M] (Y{n})) = (Y{n}) if X =/= Y .

eq ([shift X] (X{m})) = (X{suc(m)}) .

ceq ([shift X] (Y{n})) = (Y{n}) if X =/= Y .

eq ([lift X S] (X{0})) = (X{0}) .

eq ([lift X S] (X{suc(m)})) = [shift X] (S (X{m})) .

ceq ([lift X S] (Y{m})) = [shift X] (S (Y{m})) if X =/= Y .

eq S (M N) = (S M) (S N) .

eq S ([X] M) = [X] ([lift X S] M) .

Now the B-rule is given by

eq (([X] M) N) = [X := N] M . *** (B)

As an example we define the standard combinators:

op I K S : -> Trm .

eq I = [’z] ’z{0} .

eq K = [’u] [’v] ’u{0} .

eq S = [’x] [’y] [’z] ((’x{0} ’z{0})(’y{0} ’z{0})) .

The fact that (S K K) == I can be verified by reduction:

red (S K K) .

--- rewrites: 47

--- result Trm: [’z]’z{0}

Theorem 5.1 The rewrite relation induced by the above specification CINNIλ+
B is confluent.

In λ-calculus with the standard named notation, the best confluence result
that is possible is confluence modulo α-conversion, since weird renaming has to
be compensated by a notion of equivalence weaker than identity. The fact that
the previous theorem states confluence literally, i.e. in its strongest conceivable
form, is noteworthy and is made possible by the canonical treatment of names
in the CINNI calculus.

5.2 Object-Orientation: Sigma-Calculus

As another application of CINNI we give a first-order representation of the
ς-calculus [1]. Just as the λ-calculus can be seen as the most basic model
for higher-order functional programming, the object-calculus provides a basic
model for object-oriented programming. In contrast to [23] which presents
a version of the ς-calculus with explicit substitutions that is based on the
standard named representation with α-equality, the following specification is
first-order and immediately executable using Maude.

In the ς-calculus an object is seen as a set of attributes, where each attribute
has a label and a method. The labels are required to be unique inside an
object, since they are used to invoke and update the object’s attributes. Below
labels, attributes, sets of attributes, methods, objects, and object variables
are represented as elements of sorts Lab, Attr, Attrs, Meth, Obj, and Var,
respectively.

sorts Obj Attr Attrs Meth Lab Var .

var L : Lab . vars O B O’ : Obj . vars M M’ : Meth .

var A : Attr . var AA : Attrs . vars X Y Z : Qid .

op _=_ : Lab Meth -> Attr .

op emptyAttrs : -> Attrs .

subsort Attr < Attrs .

op _,_ : Attrs Attrs -> Attrs [assoc comm id: emptyAttrs].

op _{_} : Qid Nat -> Var .

subsorts Var < Obj .

op {_} : Attrs -> Obj .

op method[_]_ : Qid Obj -> Meth .

op _._ : Obj Lab -> Obj .

op _._:=_ : Obj Lab Meth -> Obj .

Notice that in contrast to the λ-calculus we make use of structural equations
(expressed by operator attributes in Maude) in the definition of the syntax of
the ς-calculus.

The last two operators are invokation and update, written as O . L and
O . L := M, respectively. In order to define these operations we need a notion
of substitution. So we instantiate the CINNI calculus to obtain the following
specification of CINNIς :

sort Subst . var S : Subst . vars n m : Nat .

op [_:=_] : Qid Obj -> Subst .

op [shift_] : Qid -> Subst .

op [lift__] : Qid Subst -> Subst .

op __ : Subst Obj -> Obj . op __ : Subst Meth -> Meth .

op __ : Subst Attr -> Attr . op __ : Subst Attrs -> Attrs .

eq ([X := O] (X{0})) = O .

eq ([X := O] (X{suc(m)})) = (X{m}) .

ceq ([X := O] (Y{n})) = (Y{n}) if X =/= Y .

eq ([shift X] (X{m})) = (X{suc(m)}) .

ceq ([shift X] (Y{n})) = (Y{n}) if X =/= Y .

eq ([lift X S] (X{0})) = (X{0}) .

eq ([lift X S] (X{suc(m)})) = [shift X] (S (X{m})) .

ceq ([lift X S] (Y{m})) = [shift X] (S (Y{m})) if X =/= Y .

eq S (L = M) = (L = S M) .

eq S emptyAttrs = emptyAttrs .

ceq S (AA, AA’) = (S AA), (S AA’) if

AA =/= emptyAttrs and AA’ =/= emptyAttrs .

eq S ({AA}) = {S AA} .

eq S (method [X] B) = method [X] ([lift X S] B) .

eq S (O . L) = (S O) . L .

eq S (O . L := M) = (S O) . L := S M .

Notice that __ is overloaded, i.e. we have a substitution application operator
for each syntactic kind. As a slight optimization we eliminated the applica-
tion of substitutions to labels. Another noteworthy point is that we added
a condition to the syntax-specific equation for application of substitutions to
attribute sets in order to avoid nontermination in the operational semantics.

Now method update (MU) and method invokation (MI) can be defined as
follows:

eq {L = M, AA} . L := M’ = {L = M’, AA} . *** (MU)

eq {L = method [X] B, AA} . L =

[X := {L = method [X] B, AA}] B . *** (MI)

Theorem 5.2 The rewrite relation induced by the above specification CINNIς+
MI+MU is confluent modulo the structural equations.

5.3 Mobile Processes: Pi-Calculus

Quite different from the λ-calculus and the ς-calculus is the π-calculus [29], a
calculus of communicating mobile processes. Here the mobility refers to the
fact that processes can exchange names of channels and use them for subse-
quent communications so that the logical communication topology can evolve
dynamically. In the λ-calculus and the ς-calculus the user is mainly interested
in the result of evaluation. Since both calculi are confluent, such a result is
unique if it exists and can be found by reduction. In the π-calculus a term is
a collection of possibly interconnected processes, and as a particular case of

a reactive system the overall dynamic behaviour is relevant. Typically such
systems are nonterminating and nondeterministic, and the states that such a
system can reach should be clearly distinguished from each other rather than
being identified by equations. So instead of using just membership equational
logic as in the λ-calculus and the ς-calculus, the capabilities of rewriting logic
to specify dynamic systems are exploited in the present example. Neverthe-
less, the equational part, which includes in particular the equationally defined
notion of substitution and the process congruence of the π-calculus, will still
play a major role.

The π-calculus distinguishes between channels and process terms which we
represent by elements of the sorts Chan and Trm, respectively. There is an
associative, commutative parallel composition P|Q defined on process terms
with the empty process nil as identity element. Given a process P, the term
out CX < CY > . P represents a process that sends the channel (name) CY

via the channel CX and then continues like P. Notice that this is not a binding
construct, whereas the construct in CX [Y] P binds the name Y in P. It repre-
sents a process that receives a channel name via channel CX and then behaves
like P with the channel variable Y (that is Y{0}) substituted by the received
channel name. Another binding construct is new [X] P which declares X to
be a local channel w.r.t. P and is also called the hiding construct. The full π-
calculus has additional constructs for choice and replication, but the fragment
introduced here will be sufficient to explain the application of CINNI in this
context. The syntax of this fragment is given by the following specification:

sorts Chan Trm .

op _{_} : Qid Nat -> Chan .

op nil : -> Trm .

op _|_ : Trm Trm -> Trm [assoc comm id: nil] .

op new[_]_ : Qid Trm -> Trm .

op out_<_>._ : Chan Chan Trm -> Trm .

op in_[_]._ : Chan Qid Trm -> Trm .

vars X Y Z : Qid . vars CX CY CZ : Chan . vars M N P Q : Trm .

The instantiation of CINNI for the π-calculus syntax, that is CINNIπ, is given
next. Since the π-calculus variables can only range over channels, it is sufficient
to have substitutions of variables by channels. As in the specification of the
ς-calculus, __ is overloaded, so that substitutions can operate on channels and
on process terms.

sort Subst . var S : Subst . vars n m : Nat .

op [_:=_] : Qid Chan -> Subst .

op [shift_] : Qid -> Subst .

op [lift__] : Qid Subst -> Subst .

op __ : Subst Chan -> Chan . op __ : Subst Trm -> Trm .

eq ([X := CZ] (X{0})) = CZ .

eq ([X := CZ] (X{suc(m)})) = (X{m}) .

ceq ([X := CZ] (Y{n})) = (Y{n}) if X =/= Y .

eq ([shift X] (X{m})) = (X{suc(m)}) .

ceq ([shift X] (Y{n})) = (Y{n}) if X =/= Y .

eq ([lift X S] (X{0})) = (X{0}) .

eq ([lift X S] (X{suc(m)})) = [shift X] (S (X{m})) .

ceq ([lift X S] (Y{m})) = [shift X] (S (Y{m})) if X =/= Y .

eq S nil = nil .

ceq S (M | N) = (S M) | (S N) if N =/= nil and M =/= nil .

eq S (out CX < CZ > . M) = out (S CX) < S CZ > . (S M) .

eq S (in CX [ Y ] . M) = in (S CX) [ Y ] . ([lift Y S] M) .

eq S (new [X] M) = new [X] ([lift X S] M) .

The process congruence is generated by the structural equations for nil and
_|_ given above and the following equations NEW1, NEW2 and NEW3 in-
volving the hiding construct. NEW2 and NEW3 are constrained by conditions
to avoid nontermination. Here we presuppose a total order _<_ on names.

eq new [X] nil = nil . *** (NEW1)

ceq (new [X] P) | Q = new [X] (P | [shift X] Q)

if P =/= nil and Q =/= nil . *** (NEW2)

ceq new [X] new [Y] P = new [Y] new [X] P if Y < X . *** (NEW3)

This completes the part of the specification that defines the process congru-
ence. As stated by the following theorem the specification is confluent, and it
is therefore not only logically but also operationally appropriate to use equa-
tions.

Theorem 5.3 The rewrite relation induced by the above specification CINNIπ+
NEW1+ NEW2+ NEW3 is confluent modulo the structural equations.

Now communication of two parallel processes via a channel CX can be ex-
pressed by the following rule that models an atomic interaction in which
(out CX < CZ > . P) sends the channel name CZ to (in CX [ Y ] . Q).

rl [communicate] : (out CX < CZ > . P) | (in CX [ Y ] . Q) =>

P | [Y := CZ] Q .

By restricting the application of the compatibility rule of rewriting logic
to the process constructors nil, _|_ and new[_]_ we make sure that, in
conformance with the π-calculus, communication never takes place inside
(out CX < CZ > . P) or (in CX [ Y ] . Q). In Maude such nonstandard
congruence properties are reflected operationally by using a suitable execution
strategy.

Related approaches to representing the π-calculus in rewriting logic, that make
use of de Bruijn index based substitution calculi, are given in [39] and [21]. In

fact, the former is more closely related to the presentation given before, since
it uses a π-calculus version of λυ so that the representation of syntax and the
substitution subcalculus arise as a special case of CINNI in the sense explained
earlier. On the other hand [39] covers the full π-calculus with choice and
replication, and uses a representation of the operational semantics exploiting
rules and strategies. A CINNI version of the full π-calculus can be obtained
by a straightforward adaptation of these rules.

5.4 Further Applications

Another application of CINNI that should be placed in the context of higher-
order logic and type theory is presented in [36], where CINNI is instantiated
to the family of pure type systems [5,38]. Pure type systems generalize the
λ-cube [3] and are considered to be of key importance, since their generality
and simplicity makes them an ideal basis for representing and implementing
higher-order logics.

Last but not least we would like to point out that the CINNI calculus is
currently being applied in the design and implementation of a proof assistant
for OCC, the open calculus of constructions, an extension of the calculus
of constructions [13] that incorporates equational logic as a computational
sublanguage. OCC supports conditional equations and conditional assertions
together with an operational semantics based on conditional rewriting modulo
equations. In fact, we have developed an experimental Maude specification of
OCC that makes use of Maude’s reflective capabilities to evaluate higher-order
equational specifications and programs with reasonable efficiency.

6 Three Orthogonal Research Directions

To place our work into the context of research conducted by other authors
we describe in the following what could be visualized as a (partial) cube of
explicit substitution calculi, namely an informal classification of three orthog-
onal research directions concerned with explicit substitution calculi and their
(potential) combinations. We restrict our attention here to first-order calculi.
We consider the minimal and well-investigated substitution calculus λυ as a
reference point and we distinguish three orthogonal directions of research:

(i) Generalizing the object language by metavariables to represent not only
closed but also open terms (enrichment).

(ii) Generalizing the fixed syntax and computation rules of the object lan-
guage from λ-calculus to languages with arbitrary syntax and computa-
tion rules (parameterization).

(iii) Generalizing the underlying representation based on de Bruijn’s indices to
Berkling’s representation allowing for an explicit representation of names
(enrichment).

Although we are not concerned with the direction (1) in this paper, this was
historically the first direction investigated. The idea to deal with open terms,
i.e. terms with metavariables, is already present in λσ [2] and λ⇑ [15] from
which λυ can be derived as a subcalculus [25]. Conversely, λυ can be extended
by a composition operator and corresponding rules to obtain λ⇑. Our main
motivation for using λυ as a starting point is its minimality and its logical
completeness in the sense that standard properties of composition are induc-
tive consequences. By definition the terms of λυ (and CINNI) do not contain
metavariables, hence confluence means ground-confluence in this context. If
terms with metavariables are considered, confluence does not hold anymore in
the core calculus. The extension λ⇑ is confluent on open terms, but does not
preserve strong normalization (even on closed terms), a property that has,
however, been established for λυ [26,4]. On the other hand, confluence on
open terms is not needed for many important applications such as execution
(see Section 5) or type checking (see [36]). This motivates our choice to use
the minimal calulus λυ, which enjoys good metatheoretic properties on terms
without metavariables, as a reference point in this paper.

Direction (2) has been pursued first in the context of combinatory reduction
systems [9] and later as a generalization of λ⇑ in [33]. The work [9] uses
combinatory reduction systems which are not first-order and belong to a level
of abstraction higher than the one we deal with in this paper. The first-
order approach presented in [33] is interesting and closely related to our work
in the sense that it investigates a condition similar to what we called well-
formedness. However, the author does not aim at preservation of confluence
results in the sense we presented them in this paper. Instead, he investigates
a number of sufficient conditions to ensure confluence. Also preservation of
strong normalization, which we consider as an important future extension of
our work, is impossible in [33] since already λ⇑ does not have this property.
Another important point is that [33] does not consider explicit names. Loosely
speaking, [33] investigates the directions (1) and (2) whereas we deal with the
directions (2) and (3). Of course, an interesting question is whether (1), (2)
and (3) can still be combined in a reasonable way if we disregard the problem
of strong normalization.

Direction (3) appears to be a natural direction in the context of explicit sub-
stitution calculi that has not been explored so far. Instead of de Bruijn’s
representation it makes use of Berkling’s representation [6,7] as a basis of an
explicit substitution calculus. In view of the clear advantages, it is surprising
that the CINNI calculus seems to be the first explicit substitution calculus
based on this representation. It might appear that the generalization is rather
straightforward, but this is definitely not true for the metatheory as indicated
by our strong normalization result which cannot be proved by just reusing the
techniques of [4,26]. This indicates that CINNI is a nontrivial generalization
that deserves a careful study. Furthermore, we think that the difference be-
comes even more challenging if other calculi such as λσ and λ⇑ are used as a

starting point for the direction (3).

7 Conclusions

The main contribution of this paper is the introduction of a generic first-order
calculus of explicit substitutions that combines the advantages of both named
and indexed notation in a natural way. The fact that our approach contains
λυ as a special case is of great help for developing the theory, since most of the
statements and proofs of [26,4] can be fruitfully generalized to our setting. In
the future we plan to further exploit this connection: As we did for confluence,
we will try to generalize and modularize the proof of preservation of strong
normalization given in [4] in a way that allows us to deduce preservation of
strong normalization for object languages different from λ-calculus. Another
interesting challenge is to extend CINNI by a notion of composition, although
it appears that composition is not compatible with preservation of strong
normalization even for indexed-based calculi.

